The Influence of Body Composition on Ethanol Pharmacokinetics using a Rate Dependent Extraction Model

Nick Holford (1), Yu Jiang (2), Daryl J. Murry (2), Timothy L. Brown(3), Gary Milavetz (2)

(1) Department of Pharmacology and Clinical Pharmacology, University of Auckland, New Zealand (2) College of Pharmacy, University of Iowa, USA (3) National Advanced Driving Simulator, Center for Computer Aided Design, University of Iowa, USA

Objectives

1) To apply an absorption rate dependent extraction model [1] to characterize the disposition of ethanol
2) To explore the effect of body composition on ethanol disposition parameters

Methods

Dosing and observations

Subjects consumed 40% ethanol v/v vodka drinks over 30 minutes on two occasions with sequence randomized doses calculated using predicted body water based on TBW, height and sex to achieve a target peak blood ethanol concentration of 650 mg/L and 1150 mg/L

- 6025 breath sample measurements were obtained from 108 subjects studied on 2 occasions
- Breath ethanol concentration was converted into blood ethanol concentration (BEC) by applying a blood: breath ratio of 2100:1
- Sampling was stopped when BEC fell below 300 mg/L

Size metrics

- Total body weight (TBW)
- Fat free mass (FFM) [2]
- Normal fat mass (NFM) [3]
  - NFM=FFM+Ffat*(TBW-FFM)
  - Ffat is a drug specific parameter that quantifies the relative contribution of fat to allometric size relative to FFM
  - Ffat was estimated separately for size related parameters

	Median	2.5-97.5% Quantile	Standard Deviation
TBW (kg)	83	54-126.6	20.4
BMI (kg/m^2)	28.5	20.4-40.2	5.7
FFM (kg)	55.7	36.7-77.7	12.9

Pharmacokinetic model

- A semi-mechanistic rate dependent extraction model with zero-order input to the gut with subsequent first order absorption was used to describe the data
- The change of hepatic first pass extraction ratio with absorption rate was accounted for by assuming a value for portal vein blood flow (Qpv) [4]
- Predicted concentration in the hepatic vein (Chv) was used as the concentration that drives mixed-order elimination
- Hepatic mixed-order (VM, Km), first order (CLFO) and non-hepatic first order elimination (CLNH) processes were evaluated with (CLFO, CLNH) and without rate dependent extraction (simple mixed order plus CLFO)
- Intrinsic hepatic clearance (CLi) was predicted by solving a quadratic function [5]

Estimation and model selection

- Data were analyzed using NONMEM 7.3.0 (ADVAN13NSIG=3, SIGL=9, TOL=9)
- Between subject variability (BSV) and between occasion variability (BOV) were tested on all parameters
- The likelihood of censored observations was predicted using the last observed BEC as the "lower limit of quantitation" and Beal's M3 method[6]
- Model selection was based on changes in objective function value (OFV)

Results and Conclusions

- OFV improved by 362 with rate dependent extraction compared to simple mixed order plus CLFO
- OFV worsened by 26.2 when Qpv predicted with TBW compared to FFM
- OFV worsened by 177.9 when V predicted with TBW compared to NFM
- A rate dependent extraction model improves model fitting compared with a simple mixed order model
- Fat free mass was the best size descriptor for Qpv
- Normal fat mass was the best size descriptor for V and total body weight for maximum elimination rate (VM)
- Predicted variability is greater than original observations compared with a fit based on simulated observations using Pred Corrected VPC

Bootstrap Parameter Estimates

Parameter	Units	Popln Estimate (RSE%)	Bootstrap (95% CI)	BSV estimate (RSE%)	BOV estimate (RSE%)
Zero-order input duration to gut D1	h	0.301 (14.7%)	(0.238, 0.413)	-	0.492 (14%)
First-order absorption from gut KA	1/h	8.83 (7.7%)	(7.70, 10.7)	-	1.21(13%)
Volume of distribution V	L/70kg NFM	38.6 (7.5%)	(35.5, 45.0)	0.091 (23%)	0.149 (21%)
Max Elim Rate VM	g/h/70kg TBW	15.8 (12%)	(11.2,18.8)	0.258 (19%)	0.309 (14%)
Conc at 50% VM Km	mg/L blood	62.5 (25%)	(45.7,92.4)	1.22 (16%)	0.438 (18%)
Qpv	L/h/70kg FFM	53.2 FIXED	-	0.166 (17%)	0.835 (12%)
CL non-hepatic CLNH	L/h/70kg TBW	0.013 (15%)	(-2.85, 2.15)	0.241(19%)	-
Ffat for volume	-	0.458 (27%)	(0.253, 0.772)	-	-
Proportional error	-	0.047 (10%)	(0.036, 0.055)	-	-
Additive error	mg/L blood	25.5 (12%)	(20.8, 32.2)	-	-

NM-TRAN CODE

```
$EST METHOD=COND NUMERICAL SLOW LAPLACIAN
PRINT=1 MAX=9999 NSIG=3 SIGL=9

$SUB ADVAN13  TOL=9

$MODEL
   COMP=(COMP1,DEFDOSE)
   COMP=(COMP2,DEFOBS)

$PK

; Group variables
   FFAT1=FFAT_V
   FFAT2=FFAT_VM
   NFW1=FFM+FFAT1*(WT-FFM)
   NFW2=FFM+FFAT2*(WT-FFM)
   ;FFM Std 56.1 for WT=70kg HT=1.76m
   STD1=56.1+FFAT1*(70-56.1)
   STD2=56.1+FFAT2*(70-56.1)
   FSIZCL=(NFW2/STD2)**0.75

   GRPV=POP_V*(NFW1/STD1)
   GRPVM=FQFVVM*POP_VM*FSIZCL
   GRPCLFO=POP_CLFO*FSIZCL
   GRPCLNH=POP_CLNH*FSIZCL
   GRPKM=FQFVKM*POP_KM
   GRPKA=POP_KA
   GRPD1=POP_D1

;Qpv from allometric scaling of ultra-sound estimate
;of flows in ten subjects with mean TBW 54 Kg [4]
;Portal Flow 53.18 L/h/70kg
;Total Hepatic Flow 66.9 L/h/70kg

   GRPQPV=53.18*(FFM/56.1)**0.75

; Between Occasion Variability
   IF (DOS.EQ.0.05) THEN
      BOVV=BOV_V1
      BOVVM=BOV_VM1
      BOVKM=BOV_KM1
      BOVD1=BOV_D11
      BOVKA=BOV_KA1
      BOVQPV=BOV_QPV1
   ELSE ; DOS=0.1
      BOVV=BOV_V2
      BOVVM=BOV_VM2
      BOVKM=BOV_KM2
      BOVD1=BOV_D12
      BOVKA=BOV_KA2
      BOVQPV=BOV_QPV2
   ENDIF

;Individual variables
   V=GRPV*EXP(BSV_V+BOVV)
   VM=GRPVM*EXP(BSV_VM+BOVVM)
   KM=GRPKM*EXP(BSV_KM+BOVKM)
   CLFO=GRPCLFO*(1+PPV_CLFO)
   CLNH=GRPCLNH*(1+PPV_CLNH)
   QPV=GRPQPV*EXP(BSV_QPV+BOVQPV)
   D1=GRPD1*EXP(BSV_D1+BOVD1)
   KA=GRPKA*EXP(BSV_KA+BOVKA)

$DES

CC=A(2)/V
CPV=CC+KA*A(1)/QPV
AXX=KM
B=QPV*(CPV+KM)-VM - CLFO*KM
CX=-(CLFO*(CPV+KM) + VM*QPV)
;Quadratic equation for CLI with mixed &
first order elimination
;CLI=(-B+SQRT(B*B-4*AXX*CX))/(2*AXX)

;Numerically better solution to quadratic [5]
IF (B.GE.0) THEN
   SGNB=1
ELSE
   SGNB=-1
ENDIF
D=-.5*(B+SGNB*SQRT(B*B-4*AXX*CX))
IF (CX/D.GT.0) THEN
   CLI=CX/D
ELSE
   CLI=D/AXX
ENDIF

ER=CLI/(QPV+CLI)
CL=ER*QPV + CLNH
CHV=ER*CPV

DADT(1)=-KA*A(1)
DADT(2)=KA*A(1)-CL*CC

$ERROR
CP=F
PROP=CP*RUV_PROP
ADD=RUV_ADD
SD=SQRT(PROP*PROP + ADD*ADD)

IF (CENSOR.EQ.0) THEN
; Greater or equal to "LLOQ"
   F_FLAG=0
   Y=CP+SD*EPS1
ELSE
; Last observed conc is "LLOQ"
   F_FLAG=1
   Y=PHI( (LASTOB-CP)/SD )
ENDIF
```

Pred Corrected VPC Original Data LLOQ=Last observed conc



Pred Corrected VPC Simulated then Fitted Data LLOQ=0.25 g/L



References

[1] Holford, N.H.G. Complex PK/PD models--an alcoholic experience. Int J Clin Pharmacol Ther, 1997. 35(10): p. 465-8.
[2] Janmahasatian, S., Duffull, S. B., Ash, S., Ward, L. C., Byrne, N. M., & Green, B. (2005). Quantification of lean bodyweight. Clin Pharmacokinet. 2005; 44(10); 1051-1065.
[3] Anderson BJ, Holford NHG. Mechanistic basis of using body size and maturation to predict clearance in humans. Drug Metab Pharmacokinet. 2009;24(1):25-36.
[4] Carlisle KM, Halliwell M, Read AE, Wells PN. Estimation of total hepatic blood flow by duplex ultrasound. Gut. 1992;33(1):92-7.
[5] Numerical Recipes in C: The Art of Scientific Computing (ISBN 0-521-43108-5) Pages 183-184 Copyright (C) 1988-1992 by Cambridge University Press.
[6] Beal SL. Ways to fit a PK model with some data below the quantification limit. Journal of Pharmacokinetics & Pharmacodynamics. 2001;28(5):481-504.